

## Simulare AO în SPICE

### Simulare PSPICE

- Simulatorul SPICE/PSPICE utilizează modele SPICE pentru componentele electrice de bază, cu ajutorul cărora realizează calculul tensiunilor și curenților în schemele electronice.
- Acesta acceptă descrierea circuitului într-un fișier text în care se specifică componentele dintre nodurile circuitului, urmată de comenzile referitoare la analizele ce urmează a fi efectuate.
- O componentă electronică se descrie pe un rând al programului și începe cu o literă specifică componentei, nodurile de conectare și valoarea.

Comenzile încep pe un rând nou cu punct în față (.). Acestea specifică fie tipurile de analize/simulări care se efectuează (**.TRAN** – simulare în timp; **.DC** – simulare în curent continuu; **.AC** – simulare în curent alternativ; **.OP** – calcul PSF, et ).

- Simularea se realizează printr-o metodă iterativă care converge către o soluție cu o eroare setată.

Exemplu de comenzi pentru analiză:

- *In timp (tranzitorie)*  
.TRAN {print step value} {final time}
- *In Current continuu .DC*  
.DC [LIN] {varname} {start} {end} {incr}  
.DC [OCT][DEC] {varname} {start} {end} {points}
- *In Current alternative .AC*  
.AC [LIN][OCT][DEC] {points} {start} {end}


Exemplu de componente care au modele SPICE și litera cu care începe descrierea acestora :

R - resistor  
C – Condensator  
L – inductor  
D – diodă  
Q – tranzistor bipolar  
J – Tranzistor TEC-J (JFET)  
M – tranzistor TEC-MOS (MOSFET)  
K – Bobine cuplate  
V – sursă independentă de tensiune  
I – sursă independentă de curent  
X - subcircuit

1. AO inversor  $A = -\frac{R_2}{R_1}$


$$U_{out} = -\frac{R_2}{R_1} U_i$$

Schema electrică în Capture/Schematic.



Programul Spice/Pspice pentru amplificatorul inversor din figura anterioară cu analiză în domeniul timp:


```
*Amplificator inversor
R1 in 2 1k
R2 2 out 4k
V2 3 0 15V
V3 0 4 15V
Vi in 0 Ac=2 sin(0 2 200)
X1 0 2 3 4 out LM741
.LIB opamp.lib
.TRAN 1u 15m
.PROBE
.END
```



2. AO neinversor  $A = 1 + \frac{R_2}{R_1}$

$$U_{out} = \left(1 + \frac{R_2}{R_1}\right) U_i$$

Schema electrică în Capture/Schematic.



\*Amplificator neinvertor

R1 0 2 1k

R2 2 out 4K

V2 3 0 15V

V3 0 4 15V

Vi in 0 Ac=2 sin(0 2 200)

X1 in 2 3 4 out LM741

.LIB opamp.lib

\*Analiza in timp

.TRAN 1u 15m

\*Analiza in curent continuu – caracteristica de transfer


.DC Vi -10V 10V 1mV

\*Analiza in curent alternativ – caracteristica de frecventa


.AC DEC 100 1Hz 100Meg

.PROBE


.END




Simulare în timp semnale intrare si iesire AO neinvertor



Caracteristica de transfer pentru AO neinvertor



Caracteristica de frecventa a amplificatorului neinvertor





Caracteristica de frecventa a amplificatorului neinvertor - Amplificarea in decibeli si banda de frecvență

### 3. AO integrator

$$U_{out} = -\frac{1}{RC} \int U_i dt$$

Schema electrică în Capture/Schematic.



\*Amplificator integrator

R in 2 1k

C 2 out 100n

V2 3 0 15V

V3 0 4 15V

Vi in 0 Ac=0 sin(0 2 200)

X1 0 2 3 4 out LM741

.LIB opamp.lib

.TRAN 1u 15m

.PROBE

.END

*Pentru semnal dreptunghiular la intrare:*

V nod+ nod- PULSE V1 V2 toff tr tf ton Perioadă

Exemplu

Vi in 0 pulse -2V 2V 0 1n 1n 2m 4m

*Pentru semnal triunghiular la intrare:*


Exemplu

Vi in 0 pulse -2V 2V 0 2m 2m 1n 4m

#### 4. AO derivator

$$U_{out} = -RC \frac{dU_i}{dt}$$

Schema electrică în Capture/Schematic.



\*Amplificator derivator

R 2 out 1k  
C in 2 100n  
V2 3 0 15V  
V3 0 4 15V  
Vi in 0 Ac=0 sin(0 2 200)  
X1 0 2 3 4 out LM741  
.LIB opamp.lib  
.TRAN 1u 15m  
.PROBE  
.END

*Pentru semnal dreptunghiular la intrare:*

V nod+ nod- PULSE V1 V2 toff tr tf ton Perioadă

Exemplu

Vi in 0 pulse -2V 2V 0 1n 1n 2m 4m

*Pentru semnal triunghiular la intrare:*

Exemplu

Vi in 0 pulse -2V 2V 0 2m 2m 1n 4m